

Solutions to Practices

1. Producing a Running Total

```
data totalTraffic;
    set pg2.np_yearlyTraffic;
    retain totTraffic 0;
    totTraffic=totTraffic+Count;
    keep ParkName Location Count totTraffic;
    format totTraffic comma12.;
run;

/*OR*/

data totalTraffic;
    set pg2.np_yearlyTraffic;
    totTraffic+Count;
    keep ParkName Location Count totTraffic;
    format totTraffic comma12.;
run;
```

2. Producing Multiple Totals

```
data work.parktypetraffic;
    set pg2.np_yearlyTraffic;
    where ParkType in ("National Monument", "National Park");
    if ParkType = 'National Monument' then MonumentTraffic+Count;
    else ParkTraffic+Count;
    format MonumentTraffic ParkTraffic comma15.;
run;

title 'Accumulating Traffic Totals for Park Types';
proc print data=work.parktypetraffic;
    var ParkType ParkName Location Count MonumentTraffic
        ParkTraffic;
run;
title;
```

3. Determining Maximum Amounts

```
data cuyahoga_maxtraffic;
    set pg2.np_monthlyTraffic;
    where ParkName = 'Cuyahoga Valley NP';
    retain TrafficMax 0 MonthMax LocationMax;
    if Count>TrafficMax then do;
        TrafficMax=Count;
        MonthMax=Month;
        LocationMax=Location;
    end;
    format Count TrafficMax comma15.;
    keep Location Month Count TrafficMax MonthMax LocationMax;
run;
```

End of Solutions