4. Using a Conditional DO Loop

```
/* f. */
data IncreaseDayVisits;
    set pg2.np_summary;
    where Reg='NE' and DayVisits<100000;
    IncrDayVisits=DayVisits;
    Year=0;
    do until (IncrDayVisits>100000);
       Year+1;
       IncrDayVisits=IncrDayVisits*1.06;
       output;
    end;
    format IncrDayVisits comma12.;
    keep ParkName DayVisits IncrDayVisits Year;
run;

proc sort data=IncreaseDayVisits;
    by ParkName;
run;

title1 'Years Until Northeast National Monuments Exceed 100,000
Visitors';
title2 'Based on Annual Increase of 6%';
proc print data=IncreaseDayVisits label;
    label DayVisits='Current Day Visitors'
          IncrDayVisits='Increased Day Visitors';
run;
title;
```

ParkName	Number of Years
African Burial Ground National Monument	14
Booker T. Washington National Monument	25
Fort Stanwix National Monument	2

```
/* h. */
data IncreaseDayVisits;
    set pg2.np_summary;
    where Reg='NE' and DayVisits<100000;
    IncrDayVisits=DayVisits;
    Year=0;
    do until (IncrDayVisits>100000);
       Year+1;
       IncrDayVisits=IncrDayVisits*1.06;
    end;
```

```
    format IncrDayVisits comma12.;
    keep ParkName DayVisits IncrDayVisits Year;
run;

proc sort data=IncreaseDayVisits;
    by ParkName;
run;

title1 'Years Until Northeast National Monuments Exceed 100,000
Visitors';
title2 'Based on Annual Increase of 6%';
proc print data=IncreaseDayVisits label;
    label DayVisits='Current Day Visitors'
          IncrDayVisits='Increased Day Visitors';
run;
title;

/* i. */
data IncreaseDayVisits;
    set pg2.np_summary;
    where Reg='NE' and DayVisits<100000;
    IncrDayVisits=DayVisits;
    Year=0;
    do while (IncrDayVisits<=100000);
       Year+1;
       IncrDayVisits=IncrDayVisits*1.06;
    end;
    format IncrDayVisits comma12.;
    keep ParkName DayVisits IncrDayVisits Year;
run;

proc sort data=IncreaseDayVisits;
    by ParkName;
run;

title1 'Years Until Northeast National Monuments Exceed 100,000
Visitors';
title2 'Based on Annual Increase of 6%';
proc print data=IncreaseDayVisits label;
    label DayVisits='Current Day Visitors'
          IncrDayVisits='Increased Day Visitors';
run;
title;
```

5. Using an Iterative and Conditional DO Loop

```
/* c. */
data IncrExports;
    set pg2.eu_sports;
    where Year=2015 and Country='Belgium'
          and Sport_Product in ('GOLF','RACKET');
    do while (Amt_Export<=Amt_Import);
       Year+1;
       Amt_Export=Amt_Export*1.07;
       output;
    end;
    format Amt_Import Amt_Export comma12.;
run;

title 'Belgium Golf and Racket Products - 7% Increase in
Exports';
proc print data=IncrExports;
    var Sport_Product Year Amt_Import Amt_Export;
run;
title;
```

Sport_Product	Number of Years	Final Year
GOLF	14	2029
RACKET	4	2019

```
/* g. */
data IncrExports;
    set pg2.eu_sports;
    where Year=2015 and Country='Belgium'
          and Sport_Product in ('GOLF','RACKET');
    do Year=2016 to 2025 while (Amt_Export<=Amt_Import);
       Amt_Export=Amt_Export*1.07;
       output;
    end;
    format Amt_Import Amt_Export comma12.;
run;

title 'Belgium Golf and Racket Products - 7% Increase in
Exports';
proc print data=IncrExports;
    var Sport_Product Year Amt_Import Amt_Export;
run;
title;
```

Sport_Product	Number of Years	Final Year	Do Exports exceed Imports?
GOLF	10	2025	No
RACKET	4	2019	Yes

```
/* j. */
data IncrExports;
    set pg2.eu_sports;
    where Year=2015 and Country='Belgium'
          and Sport_Product in ('GOLF','RACKET');
    do Year=2016 to 2025 while (Amt_Export<=Amt_Import);
       Amt_Export=Amt_Export*1.07;
    end;
    format Amt_Import Amt_Export comma12.;
run;

title 'Belgium Golf and Racket Products - 7% Increase in
Exports';
proc print data=IncrExports;
    var Sport_Product Year Amt_Import Amt_Export;
run;
title;
```

No, the Year values do not equal the final Year values before deleting the OUTPUT statement. Output happens after the DO loop due to the implicit OUTPUT. The Year column is incremented at the bottom of the DO loop before checking the DO WHILE condition at the top of the loop.

```
/* l. */
data IncrExports;
    set pg2.eu_sports;
    where Year=2015 and Country='Belgium'
          and Sport_Product in ('GOLF','RACKET');
    do Year=2016 to 2025 while (Amt_Export<=Amt_Import);
       Amt_Export=Amt_Export*1.07;
       if Year=2025 or Amt_Export>Amt_Import then output;
    end;
    format Amt_Import Amt_Export comma12.;
run;

title 'Belgium Golf and Racket Products - 7% Increase in
Exports';
proc print data=IncrExports;
    var Sport_Product Year Amt_Import Amt_Export;
run;
title;
```

6. Controlling Execution of DO Loop Statements with CONTINUE and LEAVE

```
data storm_workdays;
    set pg2.storm_summary;
    where year(StartDate)=2015 and Name is not missing;
    Duration=EndDate-StartDate+1;
    LostWork2015=0;
    do ThisDay = StartDate to EndDate;
       /* if the current day is not in 2015, exit the DO loop */
       if year(ThisDay) ne 2015 then leave;
       /* if the current day is not a work day, skip the rest
          of the statements in the loop, and loop again*/
       if weekday(ThisDay) in (1,7) then continue;
       LostWork2015+1;
    end;
    keep Name Basin MaxWindMPH StartDate EndDate
         Duration LostWork2015;
run;

title1 'Work Days Lost in 2015 due to Storms';
title2 '(where started in 2015 and ended in 2016)';
proc print data=storm_workdays;
    where year(StartDate) ne year(EndDate);
run;
title;
```

End of Solutions