

## Solutions to Practices

### 1. Using Nested Iterative DO Loops (DATA Step with No SET Statement)

```

/* b. */
data retirement;
    do Year = 1 to 6;
        Invest+10000;
        output;
    end;
run;

title1 'Retirement Account Balance per Year';
proc print data=retirement noobs;
    format Invest dollar12.2;
run;
title;

/* d. */
data retirement;
    do Year = 1 to 6;
        Invest+10000;
        do Quarter = 1 to 4;
            Invest+(Invest*(.075/4));
        end;
        output;
    end;
run;

title1 'Retirement Account Balance per Year';
proc print data=retirement noobs;
    format Invest dollar12.2;
run;
title;

/* f. */
data retirement;
    do Year = 1 to 6;
        Invest+10000;
        do Quarter = 1 to 4;
            Invest+(Invest*(.075/4));
        end;
        output;
    end;
    drop Quarter;
run;

```

```

title1 'Retirement Account Balance per Year';
proc print data=retirement noobs;
    format Invest dollar12.2;
run;
title;

```

## 2. Using an Iterative DO Loop (DATA Step with a SET Statement)

```

/* d. */
data ForecastDayVisits;
    set pg2.np_summary;
    where Reg='PW' and Type in ('NM', 'NP');
    ForecastDV=DayVisits;
    NextYear=year(today())+1;
    do Year = NextYear to NextYear+4;
        if Type='NM' then ForecastDV=ForecastDV*1.05;
        if Type='NP' then ForecastDV=ForecastDV*1.08;
        output;
    end;
    format ForecastDV comma12.;
    label ForecastDV='Forecasted Recreational Day Visitors';
    keep ParkName DayVisits ForecastDV Year;
run;

proc sort data=ForecastDayVisits;
    by ParkName;
run;

title 'Forecast of Recreational Day Visitors for Pacific West';
proc print data=ForecastDayVisits label;
run;
title;

/* e. */
data ForecastDayVisits;
    set pg2.np_summary;
    where Reg='PW' and Type in ('NM', 'NP');
    ForecastDV=DayVisits;
    NextYear=year(today())+1;
    do Year = NextYear to NextYear+4;
        if Type='NM' then ForecastDV=ForecastDV*1.05;
        if Type='NP' then ForecastDV=ForecastDV*1.08;
        if Year=NextYear+4 then output;
    end;
    format ForecastDV comma12.;
    label ForecastDV='Forecasted Recreational Day Visitors';
    keep ParkName DayVisits ForecastDV Year;
run;

```

```
proc sort data=ForecastDayVisits;
    by ParkName;
run;

title 'Forecast of Recreational Day Visitors for Pacific West';
proc print data=ForecastDayVisits label;
run;
title;
```

### 3. Using an Iterative DO Loop with a List of Values

```
/* b. */
data IncWt;
    set sashelp.class;
    where Sex='F' and Age=13;
    do Quarter=1 to 4;
        Weight+4;
        output;
    end;
run;

title 'Projected Weight for Females (Age 13)';
proc print data=IncWt;
    var Name Weight Quarter;
run;
title;

/* c. */
data IncWt;
    set sashelp.class;
    where Sex='F' and Age=13;
    do Quarter='1st Qtr', '2nd Qtr', '3rd Qtr', '4th Qtr';
        Weight+4;
        output;
    end;
run;

title 'Projected Weight for Females (Age 13)';
proc print data=IncWt;
    var Name Weight Quarter;
run;
title;
```

**End of Solutions**