Solutions to Practices

4. Creating a Custom Format from a Table

```
data type_lookup;
    retain FmtName '$TypeFmt';
    set pg2.np_codeLookup(rename=(ParkCode=Start Type=Label));
    keep Start Label FmtName;
run;

proc format cntlin=type_lookup;
run;

title 'Traffic Statistics';
proc means data=pg2.np_monthlyTraffic maxdec=0 mean sum nonobs;
    var Count;
    class ParkCode Month;
    label ParkCode='Name';
    format ParkCode $TypeFmt.;
run;
title;
```

5. Creating a Custom Format from a Table

```
data np_lookup;
    retain FmtName '$RegLbl';
    set pg2.np_codeLookup(rename=(ParkCode=Start));
    if Region ne ' ' then Label=Region;
    else Label='Unknown';
    keep Start Label FmtName;
run;

proc format cntlin=np_lookup;
run;

data np_endanger;
    set pg2.np_species;
    where Conservation_Status='Endangered';
    Region=put(ParkCode,$RegLbl.);
run;

title 'Number of Endangered Species by Region';
proc freq data=np_endanger;
    tables Region / nocum;
run;
title;
```

6. Updating a Custom Format by Using the CNTLOUT= Option

```
/*step a*/
proc format cntlin=pg2.np_types_regions;
run;

title1 'Park Frequencies by Type';
proc freq data=pg2.np_summary;
    table Type / nocum;
    format Type $TypCode.;
run;
title;

/*step b*/
proc format cntlout=typfmtout;
    select $TypCode;
run;

/*step d*/
data typfmt_update;
    set typfmtout pg2.np_newcodes;
    keep FmtName Start Label;
    FmtName='$TypCode';
run;

/*step e*/
proc format cntlin=typfmt_update;
run;

/*step f*/
title1 'Park Frequencies by Type';
proc freq data=pg2.np_summary;
    table Type / nocum;
    format Type $TypCode.;
run;
title;
```

End of Solutions